One-loop QCD self-energy diagrams using package X

T.Becher, Oct. 15, 2020

In[1]:=
```
Needs["X`"]
```

```
Package-X v2.1.1, by Hiren H. Patel
For more information, see the guide
```

Note: $\xi = 0$ corresponds to Feynman gauge.

Fermion self energy

Numerator of the diagram

In[132]:=
```
diag = CF g^2 DiracMatrix[γ_μ, γ.(p + k) + me 𝟙, γ_ν] (-g_{μ,ν} k.k + ξ k_μ k_ν);
```

In[133]:=
```
diag2 = diag // Contract
```

Out[133]=
$$\text{CF}\, g^2\, \xi\, \text{DiracMatrix}[\gamma.k,\ \mathbb{1}\,\text{me} + \gamma.k + \gamma.p,\ \gamma.k] - \text{CF}\, g^2\, \text{DiracMatrix}[\gamma_\nu,\ \mathbb{1}\,\text{me} + \gamma.k + \gamma.p,\ \gamma_\nu]\ k.k$$

In[134]:=
```
diag3 = diag2 // FermionLineExpand
```

Out[134]=
$$\text{DiracMatrix}[\gamma.p]\left(-2\,\text{CF}\, g^2\, k.k + \text{CF}\, d\, g^2\, k.k - \text{CF}\, g^2\, \xi\, k.k\right) + \text{DiracMatrix}[\,]\left(-\text{CF}\, d\, g^2\, \text{me}\, k.k + \text{CF}\, g^2\, \text{me}\, \xi\, k.k\right) + \text{DiracMatrix}[\gamma.k]\left(-2\,\text{CF}\, g^2\, k.k + \text{CF}\, d\, g^2\, k.k + \text{CF}\, g^2\, \xi\, k.k + 2\,\text{CF}\, g^2\, \xi\, k.p\right)$$

Add denominators and loop integration

In[135]:=
```
diag4 = LoopIntegrate[diag3, k, {k, 0, 2}, {k + p, m}]
```

Out[135]=
$$\text{DiracMatrix}[\,]\left(-\text{CF}\, d\, g^2\, \text{me}\, \text{PVB}[0, 0, p.p, 0, m] + \text{CF}\, g^2\, \text{me}\, \xi\, \text{PVB}[0, 0, p.p, 0, m]\right) + \text{DiracMatrix}[\gamma.p]\big(-2\,\text{CF}\, g^2\, \text{PVB}[0, 0, p.p, 0, m] + \text{CF}\, d\, g^2\, \text{PVB}[0, 0, p.p, 0, m] - \text{CF}\, g^2\, \xi\, \text{PVB}[0, 0, p.p, 0, m] - 2\,\text{CF}\, g^2\, \text{PVB}[0, 1, p.p, 0, m] + \text{CF}\, d\, g^2\, \text{PVB}[0, 1, p.p, 0, m] + \text{CF}\, g^2\, m^2\, \xi\, \text{PVB}[0, 1, p.p, 0, m, \text{Weights} \to \{2, 1\}] - \text{CF}\, g^2\, \xi\, p.p\, \text{PVB}[0, 1, p.p, 0, m, \text{Weights} \to \{2, 1\}]\big)$$

In[150]:=
```
diag5 = LoopRefine[diag4];
```

In[137]:=
```
diag6 = diag5 // Simplify
```

Out[137]=
$$\text{CF}\, g^2 \left(\text{me}\, \text{DiracMatrix}[\,]\left(2\,(-3+\xi) + (-4+\xi)\left(\frac{1}{\epsilon} + \text{Log}\left[\frac{\mu^2}{m^2}\right]\right) + \frac{(-4+\xi)\,(-m^2 + p.p)\,\text{Log}\left[\frac{m^2}{m^2 - p.p}\right]}{p.p}\right) + (-1+\xi)\,\text{DiracMatrix}[\gamma.p]\left(-1 - \frac{1}{\epsilon} - \frac{m^2}{p.p} - \text{Log}\left[\frac{\mu^2}{m^2}\right] + \frac{(m^4 - (p.p)^2)\,\text{Log}\left[\frac{m^2}{m^2 - p.p}\right]}{(p.p)^2}\right)\right)$$

Note that the package suppresses a factor

In[138]:=
```
extra = (I E^(-EulerGamma ϵ))/(4 π)^(d/2);
```

$$\text{extra} = \frac{i\, E^{-\text{EulerGamma}\, \epsilon}}{(4\pi)^{d/2}};$$

Result for the divergent term.

The result corresponds to Σ

In[139]:=
```
divergence =
  + I (extra /. d → 4 /. ϵ → 0) Coefficient[diag6, ϵ, -1] /. g^2 → 4 π αs // Simplify
```

Out[139]=
$$\frac{\text{CF}\,\alpha s\,(-\text{me}\,(-4+\xi)\,\text{DiracMatrix}[\,] + (-1+\xi)\,\text{DiracMatrix}[\gamma.\text{p}])}{4\pi}$$

In[141]:=
$$\frac{4\pi}{\alpha s}\,\text{divergence}$$

Out[141]= $\text{CF}\,(-\text{me}\,(-4+\xi)\,\text{DiracMatrix}[\,] + (-1+\xi)\,\text{DiracMatrix}[\gamma.\text{p}])$

Gluon self energy

Note: the results correspond to:

$-i\,\Pi_{\mu\nu}$

Fermion contribution to gluon self energy

In[25]:=
```
vacuumPolFerm =
  g^2 nf TF LoopIntegrate[Spur[γ_ν, γ.k + m 𝟙, γ_μ, γ.(k + q) + m 𝟙], k, {k, m}, {k + q, m}]
```

Out[25]=
$$g^2\,\text{nf TF}\,\big(q_\mu\, q_\nu\,(8\,\text{PVB}[0,1,q.q,m,m] + 8\,\text{PVB}[0,2,q.q,m,m]) + g_{\mu,\nu}\,(-4\,\text{PVA}[0,m] + 2\,q.q\,\text{PVB}[0,0,q.q,m,m] + 8\,\text{PVB}[1,0,q.q,m,m])\big)$$

In[26]:=
```
vacuumPolFerm2 = LoopRefine[vacuumPolFerm]
```

Out[26]=
$$\left(-\frac{4\,g^2\,\text{nf TF DiscB}[q.q,m,m]\,(2\,m^2+q.q)}{3\,q.q} - \frac{4\,g^2\,\text{nf TF}\,(12\,m^2+5\,q.q)}{9\,q.q} - \frac{4}{3}\,g^2\,\text{nf TF}\left(\frac{1}{\epsilon} + \text{Log}\left[\frac{\mu^2}{m^2}\right]\right)\right) q_\mu\, q_\nu + \left(\frac{4}{3}\,g^2\,\text{nf TF DiscB}[q.q,m,m]\,(2\,m^2+q.q) + \frac{4}{9}\,g^2\,\text{nf TF}\,(12\,m^2+5\,q.q) + \frac{4}{3}\,g^2\,\text{nf TF}\,q.q\left(\frac{1}{\epsilon} + \text{Log}\left[\frac{\mu^2}{m^2}\right]\right)\right) g_{\mu,\nu}$$

In[27]:=
```
vacuumPolFerm3L = vacuumPolFerm2 // Longitudinal // LoopRefine
```

Out[27]= 0

In[28]:= `vacuumPolFerm3 = vacuumPolFerm2 // Transverse // LoopRefine // DiscExpand`

Out[28]= $\frac{4}{9} g^2 \, \text{nf TF} \left(12 m^2 + 5 q.q\right) + \frac{4}{3} g^2 \, \text{nf TF} \, q.q \left(\frac{1}{\epsilon} + \text{Log}\left[\frac{\mu^2}{m^2}\right]\right) + \frac{1}{3 q.q} 4 g^2 \, \text{nf TF} \sqrt{q.q \left(-4 m^2 + q.q\right)} \left(2 m^2 + q.q\right) \text{Log}\left[\frac{2 m^2 - q.q + \sqrt{q.q\left(-4 m^2 + q.q\right)}}{2 m^2}\right]$

Note that the package suppresses a factor

In[29]:= $\text{extra} = \frac{i \, E^{-\text{EulerGamma}\,\epsilon}}{(4\pi)^{d/2}};$

In[36]:= `vacuumPolFermD =`
`+i (extra /. d → 4 /. ϵ → 0) Coefficient[vacuumPolFerm2, ϵ, -1] /. g^2 → 4 π αs //`
`Simplify`

Out[36]= $\frac{\text{nf TF}\,\alpha s \left(q_\mu q_\nu - q.q \, g_{\mu,\nu}\right)}{3\pi}$

In[123]:= $\frac{4\pi}{\alpha s}$ `vacuumPolFermD`

Out[123]= $\frac{4}{3} \text{nf TF} \left(q_\mu q_\nu - q.q \, g_{\mu,\nu}\right)$

Ghost contribution to gluon self energy

I'm suppressing the color conservation Kronecker $\delta_{a,b}$

In[62]:= `vacuumGhost = g^2 CA LoopIntegrate[(k_μ + q_μ) k_ν, k, {k, 0}, {k + q, 0}]`

Out[62]= $\text{CA}\, g^2 \left(q_\mu q_\nu \left(\text{PVB}[0, 1, q.q, 0, 0] + \text{PVB}[0, 2, q.q, 0, 0]\right) + g_{\mu,\nu} \text{PVB}[1, 0, q.q, 0, 0]\right)$

In[63]:= `vacuumGhost2 = LoopRefine[vacuumGhost]`

Out[63]= $\left(-\frac{5 \,\text{CA}\, g^2}{18} - \frac{1}{6} \text{CA}\, g^2 \left(\frac{1}{\epsilon} + \text{Log}\left[-\frac{\mu^2}{q.q}\right]\right)\right) q_\mu q_\nu + \left(-\frac{2}{9} \text{CA}\, g^2 \, q.q - \frac{1}{12} \text{CA}\, g^2 \, q.q \left(\frac{1}{\epsilon} + \text{Log}\left[-\frac{\mu^2}{q.q}\right]\right)\right) g_{\mu,\nu}$

In[64]:= `vacuumGhost3L = vacuumGhost2 // Longitudinal // LoopRefine`

Out[64]= $-\frac{1}{2} \text{CA}\, g^2 \, q.q - \frac{1}{4} \text{CA}\, g^2 \, q.q \left(\frac{1}{\epsilon} + \text{Log}\left[-\frac{\mu^2}{q.q}\right]\right)$

In[65]:= `vacuumGhost3T = vacuumGhost2 // Transverse // LoopRefine // DiscExpand`

Out[65]= $-\frac{2}{9} \text{CA}\, g^2 \, q.q - \frac{1}{12} \text{CA}\, g^2 \, q.q \left(\frac{1}{\epsilon} + \text{Log}\left[-\frac{\mu^2}{q.q}\right]\right)$

In[66]:=
```
vacuumGhostD =
  +ⅈ (extra /. d → 4 /. ϵ → 0) Coefficient[vacuumGhost2, ϵ, -1] /. g^2 → 4 π αs // Simplify
```

Out[66]= $\frac{CA\,\alpha s\,\left(2\,q_\mu\,q_\nu + q.q\,\mathbb{g}_{\mu,\nu}\right)}{48\,\pi}$

In[122]:= $\frac{4\,\pi}{\alpha s}$ vacuumGluonD

Out[122]= $\frac{1}{12}\,CA\,\left(-2\,(11+3\,\xi)\,q_\mu\,q_\nu + (19+6\,\xi)\,q.q\,\mathbb{g}_{\mu,\nu}\right)$

Gluon contribution to gluon self energy

In[110]:=
```
feynman = {VG3[k_, p_, q_, μ_, ν_, ρ_] → I g (𝕘_{μ,ν} (k - p)_ρ + 𝕘_{ν,ρ} (p - q)_μ + 𝕘_{ρ,μ} (q - k)_ν),
   (* Color structure: -ⅈ f[a,b,c]*)
   Gα[k_, α_, β_] → (-ⅈ/(k.k)) (𝕘_{α,β} - ξ (k_α k_β)/(k.k))
  };
```

In[127]:=
```
numer =
  (k.k)^2 (k.k + 2 k.q + q.q)^2 (VG3[q, -q - k, k, μ, α1, β1] × VG3[q + k, -q, -k, α2, ν, β2] ×
          Gα[k + q, α1, α2] × Gα[k, β1, β2] /. feynman) // Simplify //
     Expand // Simplify // Contract // Simplify // Expand
```

In[112]:=
```
vacuumGluon = -(CA/2) LoopIntegrate[numer , k, {k, 0, 2}, {k + q, 0, 2}];
```

In[113]:=
```
vacuumGluon2 = LoopRefine[vacuumGluon]
```

Out[113]= $\left(\frac{1}{36}\,CA\,g^2\,\left(134-36\,\xi+9\,\xi^2\right)+\frac{1}{6}\,CA\,g^2\,(11+3\,\xi)\,\left(\frac{1}{\epsilon}+Log\left[-\frac{\mu^2}{q.q}\right]\right)\right)\,q_\mu\,q_\nu + \left(-\frac{1}{36}\,CA\,g^2\,\left(116-36\,\xi+9\,\xi^2\right)\,q.q-\frac{1}{12}\,CA\,g^2\,(19+6\,\xi)\,q.q\,\left(\frac{1}{\epsilon}+Log\left[-\frac{\mu^2}{q.q}\right]\right)\right)\,\mathbb{g}_{\mu,\nu}$

In[114]:=
```
vacuumGluon3L = vacuumGluon2 // Longitudinal // LoopRefine
```

Out[114]= $\frac{1}{2}\,CA\,g^2\,q.q+\frac{1}{4}\,CA\,g^2\,q.q\,\left(\frac{1}{\epsilon}+Log\left[-\frac{\mu^2}{q.q}\right]\right)$

In[115]:=
```
vacuumGluon3T = vacuumGluon2 // Transverse // LoopRefine // DiscExpand
```

Out[115]= $-\frac{1}{36}\,CA\,g^2\,\left(116-36\,\xi+9\,\xi^2\right)\,q.q-\frac{1}{12}\,CA\,g^2\,(19+6\,\xi)\,q.q\,\left(\frac{1}{\epsilon}+Log\left[-\frac{\mu^2}{q.q}\right]\right)$

In[116]:=
```
vacuumGluonD =
  + i (extra /. d → 4 /. ϵ → 0) Coefficient[vacuumGluon2, ϵ, -1] /. g → Sqrt[4 π αs] // Simplify
```

Out[116]=
$$\frac{CA\,\alpha s\left(-2\,(11+3\,\xi)\,q_\mu\,q_\nu+(19+6\,\xi)\,q.q\,g_{\mu,\nu}\right)}{48\,\pi}$$

In[128]:=
```
4 π/αs vacuumGluonD // Expand
```

Out[128]=
$$-\frac{11}{6}\,CA\,q_\mu\,q_\nu-\frac{1}{2}\,CA\,\xi\,q_\mu\,q_\nu+\frac{19}{12}\,CA\,q.q\,g_{\mu,\nu}+\frac{1}{2}\,CA\,\xi\,q.q\,g_{\mu,\nu}$$

Transversality

Note that the **sum** of the ghost and gluon loops are transverse

In[124]:=
```
vacuumGluon3L + vacuumGhost3L
```

Out[124]= 0

Explicitly:

In[129]:=
```
vacuumGluonD + vacuumGhostD // Simplify
```

Out[129]=
$$-\frac{CA\,\alpha s\,(10+3\,\xi)\left(q_\mu\,q_\nu-q.q\,g_{\mu,\nu}\right)}{24\,\pi}$$

In[145]:=
```
4 π/αs (vacuumGluonD + vacuumGhostD) /. ξ → 0 // Simplify
```

Out[145]=
$$-\frac{5}{3}\,CA\left(q_\mu\,q_\nu-q.q\,g_{\mu,\nu}\right)$$